

دانشگاه مازندران
دانشکده فنی و مهندسی نوشیروانی بابل

مقدمه ای بر تحلیل مدارها با

Hspice

Version W.2004.09

استاد: دکتر میار نعیمی

گردآورندگان:

امید بنکدار سخی

مانی رضائی راد



روش نصب نرم افزار

با دو بار کلیک روی فایل **Setup.exe** از پوشه نصاب **Hspice** پنجره زیر به نمایش در خواهد آمد .




در صورتیکه بدون دنبال کردن پرسشها تنها بر روی دکمه **Next** کلیک نمائید ، نرم افزار **Hspice** در آدرس **C:\Synopsys\Hspice2004.09** نصب خواهد شد .

در پایان نصب پرسشی مبنی بر راه اندازی مجدد کامپیوتر بر صفحه نمایش نمایان می شود . لطفا گزینه **Restart Later** را انتخاب نموده و بر روی دکمه **Finish** کلیک نمائید .


نرم افزار جهت راه اندازی به مجوز نیاز دارد. اطلاعات مجوز در فایل **Licence.dat** قرار دارند. جهت اجرای نرم افزار به این فایل نیاز داریم که معمولا به همراه نرم افزار ارائه می شود.

فایل **Licence.dat** را به پوشه برنامه **Hspice** که آدرس آن در صفحه گذشته داده شده است، کپی نمایید. در صورتیکه آدرس نصب را تعویض نموده اید، فایل مورد نظر را در همان شاخه قرار دهید.

از منوی **Start** بر روی **Control Panel** کلیک نموده و **System** را انتخاب نمایید. پنجره **System Properties** بر روی صفحه ظاهر می شود. بر روی **Advanced** و سپس **Environment Variables** کلیک نمایید.



همانطور که در شکل نشان داده شده است در صورتیکه نصب با موفقیت همراه باشد دو خط اول در متغیر های محیطی ویندوز قرار داده می شوند . جهت اضافه کردن آدرس فایل مجوز بر روی **New** کلیک نمائید.



در پنجره **New User Variable** ، متغیر جدیدی مطابق شکل تعریف کرده و با کلیک روی دکمه **OK** آنرا ذخیره نمائید . پس از راه اندازی مجدد ویندوز ، نرم افزار قابل اجرا می باشد .




اجرای نرم افزار :

جهت اجرای نرم افزار ، Hspui W.2004.09 را که در آدرس زیر قرار دارد اجرا
نمائید .

Start > All Programs > HSPICE W.2004.09 > Hspui W.2004.09

پنجره اصلی نرم افزار بر روی صفحه ظاهر خواهد شد :



نرم افزار Hspice جهت تحلیل مدارها تنها به یک نوع فایل با پسوند .sp نیاز دارد . این فایل ، متن برنامه ایست که مدار را به شکل قابل فهم برای نرم افزار تفسیر می کند . اطلاعات متفاوتی شامل شکل مدار ، نوع تجزیه و تحلیل مورد نیاز ، ورودی ها و خروجی ها ، همه با هم در این فایل قرار دارند و هر مدار تنها در یک فایل با پسوند .sp و نام متفاوت قرار دارد .



فایل متنی (.sp) :

فایل متنی .sp باید توسط شما به وسیله یک برنامه ویرایشگر مانند Notepad یا WordPad ایجاد گردد. پس از ایجاد فایل و نوشتن خطوط برنامه در آن، نرم افزار Hspice می تواند فایل و مدار درون آنرا تحلیل نماید. هنگامی که فایل توسط نرم افزار خوانده شد، می توان محتویات فایل را نیز از طریق پنل برنامه Hspice تغییر داد.

اولین برنامه Hspice :

1- برنامه Notepad را که در آدرس زیر قرار دارد اجرا نمایید.


Start > All Programs > Accessories > Notepad

2- از منوی فایل گزینه Save را انتخاب نمایید. در قسمت File Name، sin.sp را به عنوان نام فایل تایپ نموده و از قسمت Save as type، All Files را انتخاب نمایید. با کلیک روی دکمه Save، فایل .sp مورد نظر ساخته می شود. مهم: مقدار پیش فرض Save as type، .txt می باشد و در صورتیکه فراموش نمایید آنرا تغییر دهید، حتی در صورتیکه در نام فایل SP قرار داشته باشد، Notepad آنرا با پسوند .txt ذخیره می کند.

3- خطوط زیر را عیناً در ویرایشگر وارد نموده و آنرا مجدداً ذخیره نمایید.

```
*sin.sp      sinusoidal source
.options    post node
v          1  0      sin( 0  1  100meg  2n  5e7  0)
r          1  0      1
.tran      .05n 50n
.end
```

4- در برنامه Hspice بر روی گزینه Open کلیک نموده و فایل sin.sp را باز نمائید




5- برای تحلیل برنامه بر روی دکمه Simulate کلیک نمائید . پس از هر بار تحلیل ، Hspice ، فایل با نام فایل .sp و پسوند .lis تولید می کند که در آن گزارشی از تحلیل قرار داده شده است . جهت دیدن این گزارش بر روی Edit LL کلیک نمائید . در انتهای گزارش در صورتیکه تحلیل با موفقیت همراه باشد ، جمله Job Concluded دیده خواهد شد . گزارش را ببندید .

```
Opening plot unit= 79
file=c:\synopsys\hspice2004.09\demo\hspice\sources\sin.tr0

***** job concluded
1 ***** HSPICE -- W-2004.09 (20040730) 18:14:48 04/26/2006 pent
*****
*sin.sp      sinusoidal source
***** job statistics summary          tnom= 25.000 temp= 25.000
*****
```

6- جهت دیدن شکل موج های خروجی بر روی Avanwaves کلیک نمائید . دو پنجره Results Browser و Avanwaves ظاهر خواهند شد . از پنجره Results Browser بر روی Sinusoidal : *sin.sp و سپس در قسمت Types بر روی Voltages در قسمت Curves بر روی 1 کلیک نمائید .

شکل موج منبع ولتاژ تعریف شده در فایل sin.sp در پنجره Avanwaves نمایش داده میشود .



فرمت نوشتن فایل .sp :

به غیر از اولین سطر برنامه که شامل عنوان آن می باشد . ترتیب سطر ها در اجرای برنامه هیچ تاثیری ندارند . با این حال معمولاً هر نویسنده ترتیبی را در نوشتن فایلها رعایت مینماید تا تغییر و اصلاح آنها به راحتی صورت گیرد .
فایلهای برنامه Hspice را به طور کل می توان به قسمتهای زیر تقسیم کرد :

Title

اولین سطر فایل یک عنوان اختیاری در رابطه با مدار و معرفی آن است . این سطر هیچگونه تأثیری در نتیجه تحلیل برنامه نخواهد داشت . به عبارتی این سطر کلاً توسط برنامه نادیده گرفته می شود . مانند سطر اول در برنامه ی ما :

```
*sin.sp sinusoidal source
```


Library

گاهاً در مدارها ، هنگام استفاده از بعضی المان ها ، پارامترهای المان در فایل های جداگانه ای نهفته می باشد . به عنوان مثال در صورت استفاده از ترانزیستور در مدار ، برای هر ترانزیستور خاص پارامترهایی از قبیل مقادیر خازنهای داخلی و بسیاری از پارامترهای مرتبط در درون فایل قرار داده می شوند . جهت استفاده از این ترانزیستور در مدار ، برنامه بدون وجود این فایل مرجع قابل اجرا نمی باشد . دیود ها نیز نمونه دیگری از اینگونه المانها میتوانند باشند .

Parameters

گاهاً در برنامه می توان عدد ثابتی را به یک حرف یا کلمه نسبت داد . اینگونه تعاریف را به عنوان پارامتر در مدار تعریف می نمائیم . مانند :

$$\text{.param c1} = 10\text{nF}$$

Circuit Description

در این قسمت توضیحات و بدنه اصلی مدار نوشته می شود که می تواند شامل المانهای مدار ، منابع و چگونگی ارتباط آنها باشد . به عنوان مثال خطوط زیر را در نظر بگیرید :

```
v 1 0 sin(0 1 100meg 2n 5e7 0)
r 1 0 1
```

خط اول نشان می دهد یک منبع سینوسی به نام v وجود دارد به گونه ای که سر مثبت آن به گره ای به نام 1 و سر منفی آن به گره ای به نام صفر یا زمین مدار متصل است . پارامترهای داخل پرانتز متعلق به منبع سینوسی بوده و در ادامه در باره آنها صحبت خواهد شد . خط دوم مقاومتی به نام r را با مقدار 1 اهم نشان می دهد که موازی منبع سینوسی قرار دارد .

Analysis

شامل همه دستوراتی است که برای برنامه Hspice ، نوع تحلیل و نحوه خروج داده ها را نشان می دهد و می تواند شامل دستورات زیادی مانند op ، probe ، print ، tran و ... باشد . به عنوان مثال خط 50n 5n .tran از Hspice می خواهد تا مدار را برای 50 نانوثانیه و به فاصله های نیم نانو تحلیل زمانی حالت گذرا نماید .

End

خط آخر هر برنامه نیز برابر **end** خواهد بود که نشان دهنده پایان برنامه است .


مقدمه ای بر برنامه نویسی با Hspice

تخصیص شماره یا اسم به گره های مدار :

هر مدار شامل چند گره و چند شاخه و یک یا چند ورودی و یا خروجی خواهد بود . گره ها به دو صورت مشخص می شوند . با نام یا با شماره . در نامگذاری گره های غیر عددی ، نام گره ها نباید از 16 کاراکتر بیشتر باشد و باید حتما با یکی از حروف الفبا آغاز گردد . در نامگذاری می توان از کاراکترهایی مانند ! # \$ % * + - / < > [] استفاده کرد و در ضمن هیچ تفاوتی بین حروف کوچک و بزرگ وجود ندارد .

گره های مدار در برنامه Hspice از شماره صفر تا 1E16-1 شماره گذاری می شوند . گره صفر و GND و GND! و GRONND همگی نشان دهنده گره زمین هستند . ترتیب شماره گذاری بقیه گره ها از اهمیت برخوردار نیست . در شماره گذاری گره ها می توان به جای اعداد از اسمهای مختلفی جهت معنی دار شدن اسم آنها استفاده نمود .

تمام گره ها باید از طریق یک مسیر DC به زمین متصل شوند . این مسیر می تواند شامل مقاومت ها ، منابع ، سلف ها ، دیود و یا ترانزیستور تأمین گردد . در صورتیکه در یک مدار گره ای وجود دارد که با هیچ مسیر DC به زمین متصل نشده است ، می توان با اضافه نمودن یک مقاومت بزرگ از آن گره تا زمین این مسیر را ایجاد کرد .



همانطورکه در شکل نشان داده شده است ، گره 4 هیچ مسیر DC به زمین ندارد . لذا از

یک مقاومت بزرگ بین گره 4 تا زمین استفاده می نمائیم .

مقادیر مجاز المانها و پارامترهای آنها :

مقادیر المان های مدار در نرم افزار Hspice می توانند اعداد صحیح مثبت و منفی ، اعشاری و نمایی را شامل شوند .

می توان از حروف لاتین بعنوان پسوند برای اعداد جهت بیان ضریب عدد استفاده نمود . جدول زیر را مشاهده نمائید :

A	1e-18	K	1e3
F	1e-15	MEG (or X)	1e6
P	1e-12	MI	25.4e6
N	1e-9	G	1e9
U	1e-6	T	1e12
M	1e-3		

جهت نشان دادن واحدها از حروف زیر استفاده می شود :

A	آمپر	H	هانری
V	ولت	F	فاراد
Hz	هرتز	DEG	درجه
OHM	اهم		

روش بیان المانها:

در برنامه Hspice هر نوع عنصر یا المان مدار دارای یک اسم است و این اسم بسته به نوع المان با یک حرف خاص شروع می شود. نام المانها می توانند تا 10 کاراکتر ادامه یابند. جدول زیر اولین حرف در نامگذاری اجزای مدار برای المان های مختلف را نشان می دهد:

B	بافر	K	ضریب کوپلاژ
C	خازن	L	سلف
D	دیود	M	ترانزیستور ماسفت
E	منبع ولتاژ وابسته به ولتاژ	Q	ترانزیستور BJT
F	منبع جریان وابسته به جریان	R	مقاومت
G	منبع جریان وابسته به ولتاژ	T, U, W	خطوط انتقال
H	منبع ولتاژ وابسته به جریان	V	منبع ولتاژ مستقل
I	منبع جریان مستقل	X	فراخوانی زیر مدار
J	ترانزیستور جی فت و مسفت		

نام هر المان می تواند دارای 10 کاراکتر مختلف باشد و از این تعداد کاراکتر حرف ابتدایی حتماً باید مطابق جدول و با توجه به نوع المان انتخاب گردد.



معرفی المانهای دو سر :

بطور کلی فرمت بیان المانهای دو سر از یک مدار در برنامه Hspice به فرم زیر است :

مقدار	N-	N+	نام المان
-------	----	----	-----------

N+ و N- شماره یا نام دو گره ای هستند که المان بین آن دو قرار گرفته است و قرارداد بر اینست که جریان از گره N+ به سمت گره N- جاری است.

به عنوان چند مثال :

R1	1	2	10KOHM
C2	2	3	2.2nF
L3	3	4	1mH
V1	1	0	10V

بیان مدل برای المان :

گاهی مقدار یک المان ، شامل مجموعی از پارامترهای مختلف می باشد . در این صورت فرمت کلی بیان المان متفاوت بوده و به صورت زیر تعریف می شود :

[مقدار]	نام مدل	N-	N+	نام المان
-----------	---------	----	----	-----------

پارامترهای مختلف را با یک نام مدل نشان داده و سپس پارامترهای موجود در مدل را توسط دستور Model تعیین می کنند . بدین ترتیب امکان آن وجود دارد که از یک مدل برای تعداد زیادی از المانها در مدار استفاده نمود .

فرم کلی دستور مدل به شکل زیر است :

.MODEL mname type P1=X1 P2=X2 ... Pn=Xn

mname اسمی انتخابی است که به عنوان نام مدل برای آن انتخاب می شود و **type** مشخص کننده نوع آن است. **P1, P2, ...** پارامترهای مربوط به آن مدل هستند. **X1, X2, ...** به ترتیب مقادیر آن پارامترها می باشند .

AMP	تقویت کننده تفاضلی	PJF	جی فت p کانال
C	خازن	PLOT	مدل plot برای GRAPH.
CORE	هسته مغناطیسی	PMOS	ماسفت p کانال
D	دیود	PNP	ترانزیستور دوقطبی
L	هسته مغناطیسی سلف	R	مدل مقاومت
NJF	جی فت n کانال	U	مدل خط انتقال پر اتلاف (چند رشته)
NMOS	ماسفت n کانال	W	مدل خط انتقال پر اتلاف
NPN	ترانزیستور دوقطبی	SP	پارامتر S
OPT	Optimization		

به عنوان مثال :

```
D1 1 2 D914
.MODEL D914 D IS=1e-15 VJ=0.6 CJA=1.2e-13
+CJP=1.3e-14 RS=1e+8 BV=70V
```

دیود D1 بین گره های 1 و 2 قرار گرفته است . و دارای مدل D914 می باشد . Hspice به دنبال خطی خواهد گشت که مدلی برای D914 در آن تعریف شده باشد .

منابع ولتاژ و جریان در Hspice :

این منابع در Hspice به دو دسته مستقل و وابسته تقسیم می شوند . منابع مستقل و منابع وابسته . در این نسخه تنها به بررسی تعدادی از منابع مستقل می پردازیم .

منابع مستقل :

منبع ولتاژ یا جریان DC :

منبع ولتاژ و جریان DC در Hspice به صورت کلی زیر بیان می شوند :

Vx	N+	N-	DC	value	or	Vx	N+	N-	value
Ix	N+	N-	DC	value	or	Ix	N+	N-	value

N+ و N- به ترتیب سر های مثبت و منفی منبع می باشند .

به عنوان مثال **VDD VDD 0 DC 3.3V** نشان دهنده یک منبع ولتاژ DC با مقدار

ولتاژ 3.3 ولت بین گره VDD و 0 می باشد . توجه شود که نام گره می تواند با نام منبع یکی باشد و برنامه می تواند به راحتی آنها را از هم تمیز دهد .

به عنوان مثالی دیگر **Iref vdd drain 10mA** نشان دهنده یک منبع جریان با مقدار

10mA بین گره vdd و گره ای به نام drain می باشد .

منبع ولتاژ یا جریان AC :

این نوع منبع در تحلیل نویز و AC کاربرد دارد و به صورت زیر تعریف می شود :

Vx	N+	N-	AC	value	θ
Ix	N+	N-	AC	value	θ

به عنوان مثال جهت بیان $V_x = A \sin(\omega t + \theta)$ متصل به گره های 1 و زمین داریم :

Vx	1	0	AC	=	A	,	θ
Vx	1	0	AC		A		θ

تفاوتی بین دو جمله فوق وجود ندارد .

منبع ولتاژ و جریان مرکب :

اگر منبعی دارای مقدار AC و DC باشد به صورت زیر بیان می شود :

V_x	N+	N-	DC value	AC value	θ
I_x	N+	N-	DC value	AC value	θ

نمونه :

منبع ولتاژ V_x به مقدار ولتاژ $A + B\sin(\omega t + \theta)$ ولت را داریم :

V_x	1	0	DC = A	AC = B	, θ
V_x	1	0	DC A	AC B	θ

منبع ولتاژ یا جریان Sinusoidal :

این نوع از منابع جهت تجزیه و تحلیل گذرای مدارها استفاده می شوند :

V_x	N+	N-	SIN	VO	VA	FREQ	TD	α	θ
I_x	N+	N-	SIN	IO	IA	FREQ	TD	α	θ

عبارتهای بالا به ترتیب از روابط زیر گرفته شده اند :

$$V(t) = V_0 + V_A + e^{-\alpha(t-t_d)} \sin[2\pi f(t - t_d) + \theta]$$

$$I(t) = I_0 + I_A + e^{-\alpha(t-t_d)} \sin[2\pi f(t - t_d) + \theta]$$

به عنوان مثال :

Vin 1 0 Sin 0 1V 1KHz

معادل یک منبع ولتاژ سینوسی با سطح DC صفر و دامنه 1 ولت و فرکانس یک کیلوهرتز بین گره های 1 و 0 خواهد بود . از آنجایی که دیگر پارامترها تعیین نگردیده اند ، مقدار آنها توسط Hspice صفر در نظر گرفته می شود .

منبع ولتاژ یا جریان پالسی :

این نوع منابع نیز به ترتیب دارای شکل کلی زیر خواهند بود :

Vx	N+	N-	Pulse	V1	V2	TD	TR	TF	PW	PER
Ix	N+	N-	Pulse	I1	I2	TD	TR	TF	PW	PER

به عنوان مثال :

```
Vin 1 0 Pulse 2 3.3V 5uS 5nS 10nS 500uS 1mS
```

معادل منبع ولتاژ پالسی بین گره های 1 و 0 که دارای ولتاژ اولیه 2 بوده و پس از مدت زمان 5 میکرو ثانیه به سطح 3.3 ولت خواهد رسید و به مدت 500uS در این سطح باقی خواهد ماند . زمان اوج برابر 5 نانو ثانیه و زمان افت برابر 10 نانو ثانیه می باشد . این پالس هر 1mS تکرار خواهد شد .

عناصر غیر فعال :

فرم خلاصه شده تعریف عناصر در زیر گرد آورده شده است . جهت تعریف دقیق هر عنصر به راهنمای همراه نرم افزار Hspice مراجعه نمایید .

مقاومت :

```
Rxxx node1 node2 value
```

به جای xxx هر کاراکتر دیگری که نشانگر نام مقاومت باشد می تواند قرار گیرد . value مقدار مقاومت بر حسب اهم می باشد . مقاومت بین گره های node1 و node2 قرار دارد . از آنجایی که مقاومت دارای جهت نمی باشد ، تفاوتی بین ترتیب گره ها وجود ندارد .

مثال :

```
R1 1 2 10K
```



خازن :

Cxxx node1 node2 value <IC = value >

در رابطه بالا پس از نام گره ها مقدار خازن بر حسب فاراد نوشته می شود . قسمت داخل پرانتز اختیاری بوده و نشان دهنده ولتاژ اولیه خازن است که مقدار پیش فرض آن صفر در نظر گرفته می شود . باید توجه داشت که IC از عبارت **Initial Condition** گرفته شده است و ارتباطی با جریان خازن ندارد .

مثال :

```
C1 1 0 10uF
CL Vin 0 10pF IC=3V
```

سلف :

ساده ترین روش بیان سلف ها به صورت زیر است :

Lxxx node1 node2 value <IC = value>

مقدار سلف بر حسب هانری بوده و مقدار اولیه نشان دهنده جریان لحظه صفر سلف هنگام تحلیل مدار است .

مثال :

```
L1 vdd out 1mH
L2 2 0 5uH IC=1mA
```

سلف های کوپلاژ :

جهت تعریف سلف های کوپلاژ ابتدا هر دو سلف به طور معمول تعریف شده و سپس ضریب کوپلاژ طبق رابطه زیر بین آنها برقرار می شود .

Kxxx Lyyy Lzzz <K= coupling>

مثال :

```
LA 1 2 5uH
```

LB	3	4	10uH
K1	LA	LB	0.8

باید توجه داشت که سر مثبت سلف معادل سر نقطه دار آن در مدار می باشد .

به عنوان مثالی دیگر ، یک ترانسفورماتور با سر وسط و ضریب کوپلاژ 0.999 به صورت زیر

مدل می شود :

L1	1	2	0.7mH
L2	3	4	0.5mH
L3	5	6	0.4mH
K12	L1	L2	0.999
K13	L1	L3	0.999
K23	L2	L3	0.999

در صورت یکسان بودن ضریب کوپلاژ مانند بالا می توان هر سه عبارت آخر را در یک خط به

صورت خلاصه بیان نمود :

KL	L1	L2	L3	0.999
----	----	----	----	-------

معرفی و مدل سازی عناصر فعال و نیمه هادی :

1- دیود :

Dxxx nplus nminus mname <OFF>

nplus و **nminus** به ترتیب گره های آن د و کاتُ د دیود می باشند و جریان از گره اول به دوم برقرار می شود . **mname** نام مدلی است که پارامترهای دیود را تعریف می کند . برای هر مدل مقادیر پارامترها متفاوت است . مدلها می توانند توسط برنامه نویس با جزئیات کامل تعریف شوند یا اینکه این مدلها از قبل تعریف شده ای باشند که به برنامه فراخوانی می شوند . در حالت اخیر نام فایلی که مدل در آن قرار دارد در ابتدای برنامه ذکر می شود تا به برنامه فراخوانی شود .

مثالی از تعریف یک دیود و مدل آن به صورت زیر است :

D1 D+ D- DIN4001

**.Model DIN4001 D LEVEL=1 XP=0 EG=1.1 XO1=0 XOM=0 XM=0
+WP=0 WM=0 LP=0 LM=0 AF=1.0 JSW=0 PB=0.65 PHP=0.8
+M=0.2994 FC=0.95 FCS=0.95 MJSW=0.5 TT=2.446e-9 BV=4.65
+RS=19 IS=1.485e-11 CJO=1.09e-9 CJP=0 PJ=0 N=1.615 IK=0.0
+IKR=1.100e-2 IBV=2.00e-2**

توجه شود که در صورتیکه تعداد پارامترها زیاد و فضای کافی جهت نوشتن همه آنها در یک سطر وجود ندارد ، ادامه به ابتدای سطر بعد منتقل شده و در ابتدای سطر علامت + قرار میگیرد.

جهت نوشتن مدل برای عناصر اطلاع کافی از جزئیات فنی عنصر الزامی می باشد . این اطلاعات معمولاً به طور کامل توسط کارخانه سازنده در **Datasheet** قطعه ذکر شده اند . جهت مطالعه دقیق پارامترهای لازم هر مدل به کتاب مرجع نرم افزار **Hspice** مراجعه نمائید .



2- ترانزیستور BJT :

خط فرمان خلاصه شده تعریف ترانزیستور **BJT** به فرم زیر می باشد :

Qxxx nc nb ne mname <OFF>

nc ، **nb** ، **ne** به ترتیب نام گره های کلکتور ، بیس و امیتر بوده و **mname** نام مدلی است که ترانزیستور بر اساس آن تعریف می شود . در صورتیکه در ادامه از کلمه **OFF** استفاده شود، حالت اولیه ترانزیستور خاموش خواهد بود .

مثال :

Q1 node_c node_b node_e T2n2222a
.Model T2n2222a NPN LEVEL=1 ISS=0 (ادامه)

2- ترانزیستور JFET و MESFET :

Jxxx nd ng ns mname <OFF>

تعریف این ترانزیستور نیز مانند ترانزیستور BJT بوده و تنها گره ها هستند که به ترتیب نشانگر درین ، گیت و سورس می باشند .

3- ترانزیستور MOSFET :

**Mxxx nd ng ns <nb> mname <<L = >length> <<W = >width>
 + <AD = val> AS = val <PD = val> <PS = val>
 + <NRD = val> <NRS = val> <RDC = val> <RSC = val> <OFF>
 + <IC = vds,vgs,vbs> <M = val> <DTEMP = val>
 + <GEO = val> <DELVTO = val>**

مانند ترانزیستور های گروه سوم ، گره ها به ترتیب درین ، گیت ، سورس و بالک (بدنه) می باشند . **mname** نام مدلی است که برای ترانزیستور تعریف شده است . تعدادی از پارامترهای دیگر در جدول زیر تعریف شده اند .

L	طول کانال ترانزیستور (متر)
W	پهنای کانال ترانزیستور (متر)
AD	مساحت درین
PD	محیط درین با احتساب گوشه های کانال
PS	محیط سورس با احتساب گوشه های کانال
OFF	حالت اولیه ترانزیستور در لحظه صفر
DTEMP	اختلاف دمای ترانزیستور و مدار (درجه سلسیوس)


مثال :

M1 vdd gate source 0 nmos L=.35u W=10u

نشان دهنده یک مسافت با طول کانال و پهنای کانال به ترتیب 35 و 10 میکرو متر می باشد که درین آن به گره vdd و بدنه آن نیز به زمین متصل است . معمولاً در ترانزیستورها در نوع n-channel بدنه به پایین ترین ولتاژ مدار و در نوع p-channel بدنه به بالاترین ولتاژ مدار متصل می شود . در این ترانزیستور از مدل nmos استفاده شده است .

یک تمرین نمونه :

به عنوان تمرینی ساده المانهای مدار شکل زیر را با آنچه تاکنون خوانده ایم بیان می نمائیم :



Vs	1	0	sin	0	1	1KHz
R1	1	2	10K			
C1	2	3	10u			
L1	3	4	1mH			
C2	4	0	100p			
R2	4	0	100MEG			

مقادیر المانها به دلخواه انتخاب شده اند .

خط آخر مقاومتی بزرگی است که ارتباط بین گره 4 و زمین را از نظر DC برقرار می نماید .

انواع متغیرها به عنوان خروجی :

در هر مداری ولتاژها و جریانهای مختلفی را می توانیم به عنوان متغیرهای خروجی تعریف کنیم . شکل کلی متغیرهای خروجی به همراه مثالی در جدول ارائه شده است :

فرم کلی	توضیح	مثال	توضیح مثال
$V(\text{node})$	ولتاژ گره نسبت به زمین	$V(7)$	ولتاژ گره 7 نسبت به زمین
$V(\text{node1}, \text{node2})$	اختلاف ولتاژ دو گره	$V(3,4)$ or $V(3\ 4)$	اختلاف ولتاژ گره 3 و 4
$V(\text{نام ا لمان})$	اختلاف پتانسیل دو سر یک ا لمان دو سر	$V(R1)$ or $V(D1)$	
V حرف ترمینال(نام ا لمان)	ولتاژ یک سر از ا لمان سه سر نسبت به زمین	$VD(M3)$	ولتاژ درین ترانزیستور $M3$
V نام دو ترمینال(نام ا لمان)	اختلاف ولتاژ دو سر از یک عنصر سه سر	$VCE(Q4)$	ولتاژ کلکتور- امیتر ترانزیستور $Q4$
		$I(D1)$ or $I(R1)$	جریان مقاومت $R1$ یا جریان دیود $D1$
I نام ترمینال(نام ا لمان)	جریان ورودی به یک ترمینال عنصر	$IB(Q2)$ or $IC(Q1)$	

در تجزیه و تحلیل AC ممکن است اندازه ، فاز ، بخش حقیقی و یا موهومی یک متغیر مطلوب باشد . با اضافه نمودن پسوند مناسب به ولتاژها و جریانها مقادیر مطلوب بدست می آیند :

$VM(3)$	اندازه ولتاژ گره 3	$VDSP(M6)$	اندازه فاز اختلاف ولتاژ درین سورس ترانزیستور $M6$
$VM(4,2)$	اندازه اختلاف ولتاژ گره های 4 و 2	$VR(2,3)$	بخش حقیقی اختلاف ولتاژ
$VP(4,2)$	فاز اختلاف ولتاژ بین گره های 4 و 2	$VI(2,3)$	بخش موهومی اختلاف ولتاژ
$VDB(7)$	اندازه ولتاژ گره 7 بر حسب dB	$IM(D1)$	اندازه دامنه جریان در دیود $D1$
$VCM(Q2)$	اندازه ولتاژ کلکتور ترانزیستور $Q2$	$IP(D1)$	اندازه فاز جریان دیود $D1$
$IR(Vin)$	بخش حقیقی جریان منبع ولتاژ Vin		

انواع تجزیه و تحلیل مدار در برنامه Hspice :

انواع مختلف تجزیه و تحلیل مدار در برنامه Hspice قابل اجراست که مهمترین آنها

عبارتند از :

1- تجزیه و تحلیل DC

2- تجزیه و تحلیل AC

3- تجزیه و تحلیل حالت گذرا

تجزیه و تحلیل DC مدار :

تجزیه و تحلیل DC خود دارای چند نوع می باشد که در همه انواع آن ، خازنهای مدار باز و تمام سلف ها اتصال کوتاه در نظر گرفته می شوند . انواع آن عبارتند از :

الف : OP (Operating Point) :

این دستور نقاط کارمدار را بدست می آورد . این نقاط شامل ولتاژگره ها ، جریان شاخه ها و توان مصرفی عناصر می باشند . علاوه بر موارد فوق پارامتر های مدل سیگنال کوچک عناصر فعال مانند ترانزیستورها نیز ارائه می گردند . نتایج این دستور در فایلی با پسوند **.lis** که همانام با فایل مدار تحلیلی می باشد ، ذخیره شده و به وسیله یک ویرایشگر ساده قابل رویت است.

فرمهای خلاصه این دستور از قرار زیر است :

.op
.op [format] [time] [format] [time]

خط اول نقاط کار را بطور کامل گزارش می نماید . در صورتیکه فرم خاصی از تحلیل مد-

نظر باشد از خط دوم استفاده می نمائیم . پارامتر های خط دوم در جدول آمده اند :

Format	<p>هر کدام از کلمات زیر قابل استفاده اند . تنها وجود حرف اول کافی است .</p> <p>ALL : همه نقاط کار شامل ولتاژها ، جریانها ، توان و ... به طور کامل نمایش می یابند .</p> <p>BRIEF : از هر المان در مدار و ولتاژ ، جریان و توان آنها یک گزارش تک خطی ارائه می دهد .</p> <p>CURRENT : جدول ولتاژ با یک ارائه خلاصه از جریان و توان المانها</p> <p>VOLTAGE : تنها جدول ولتاژها را نمایش می دهد .</p>
Time	در صورتیکه در جلوی هر فرمت نوشته شود ، زمان آن رویداد را هم ثبت می نماید.

ب : **SENS** . (حساسیت) :

حساسیت ولتاژها و جریان های مشخصی از مدار را نسبت به تغییرات پارامتر های مدار ارائه می کند . شکل کلی دستور به فرم زیر است :

یک یا چند جریان یا ولتاژ **SENS** .

ج : **TF** . :

مقدار تابع تبدیل سیگنال کوچک ، مقاومت ورودی و مقاومت خروجی مدار را بدست می آورد . به عنوان مثال در صورتیکه **V(in)** ولتاژ ورودی و **V(out)** ولتاژ خروجی باشند . برای بیان بهره ولتاژ که در فایل ثبت می شود از دستور زیر استفاده می نمائیم :

TF V(out) V(in) .

د : **DC** . (مشخصه انتقالی) :

مشخصه انتقالی مدار را بدست می دهد . بدین مفهوم که ورودی بطور **DC** بین دو حد مشخص تغییر می کند و خروجی در هر نقطه اندازه گیری شده و سپس منحنی خروجی بر حسب تغییرات ورودی ارائه می گردد .

شکل ساده این دستور به فرم زیر است :

.DC Input Start Stop Step

پارامترهای آن به ترتیب نام متغیر ، مقدار شروع ، مقدار نهایی و مقدار گام افزایش ورودی می باشند .


به عنوان مثال برنامه زیر را در نظر بگیرید :

```
.lib 'N_WS.LIB' WS
Vs 3 0
Vg 1 0
Vd 2 0 3.5
MOSFET1 2 1 3 0 MODN L=3U W=9U
.DC Vg 0 5 0.01 Vs 0 2.5 .4
.PRINT I(MOSFET1)
.END
```

توضیح آنکه :

جهت اجرای این برنامه به فایل **N_WS.LIB** نیاز است . زیرا که مدل **MODN** در درون آن قرار دارد . هنگامیکه مانند این مثال با دو تغییر پارامتر روبرو هستیم ، محور **X** جدول توسط تغییر پارامتر اول درجه بندی می گردد و پارامتر دوم باعث تولید چندین رسم خواهد شد . در شکل هر منحنی به ازای یکی از مقادیر **Vs** رسم شده است .

خروجی جریان ماسفت در پنجره **Avanwaves** برابر زیر خواهد بود :



تجزیه و تحلیل AC (.AC) :

این تجزیه و تحلیل پاسخ فرکانسی را در یک پهنای باند مشخص از فرکانس بدست می آورد و برحسب درجه بندی شدن محور فرکانس به یکی از حالت‌های خطی، لگاریتم در مبنای 8 و لگاریتم در مبنای 10 نشان می دهد.

برای انجام این نوع تحلیل، وجود یک منبع AC یا یک منبع مرکب در مدار الزامیست. فرم کلی دستور به شکل زیر است:

```
.AC type np fstart fstop
```

به جای پارامتر **type** جهت درجه بندی شدن محور فرکانس به فرم خطی از کلمه **LIN** استفاده شده و برای درجه بندی لگاریتمی در مبنای 8 و 10 به ترتیب از کلمات **OCT** و **DEC** استفاده می شود. پارامتر **np** نشان دهنده تعداد نقاطی است که تحلیل برای آنها صورت گرفته و دو پارامتر آخر، فرکانس آغاز و پایان می باشند.

معمولاً جهت نمایش پهنای باند باریک از محور خطی و جهت نمایش پهنای باند متوسط و عریض از محورهای لگاریتمی در مبنای 8 و 10 استفاده می شود.

مثال :

```
.AC LIN 150 100Hz 300Hz  
.AC OCT 10 100Hz 10KHz  
.AC DEC 10 100Hz 10MHz
```

نکته : در محورهای لگاریتمی مقدار پارامتر **np** نشان دهنده تعداد نقاط در هر اکتاو و یا در هر دهه می باشد.

می توان با دستورهای زیر تجزیه و تحلیل را به ازای مقادیر متفاوتی از یک ورودی یا در دماهای متفاوت انجام داد :

```
.AC type np fstart fstop <SWEEP var start stop step>  
.AC type np fstart fstop <SWEEP type np start stop>
```

مثال :


```
.AC DEC 10 100 100k SWEEP Iref DEC 2 10n 10u  
.AC DEC 10 100 100k SWEEP temp 25 50 75 100
```

برنامه زیر را در نظر بگیرید :


.OPTION LIST NODE POST

```
V1 1 0 10 AC 1
R1 1 2 1K
R2 2 0 1K
C1 2 0 .001U
```

```
.AC DEC 5 1K 1MEG
.PRINT AC V(1) V(2) I(R2) I(C1)
.END
```



خروجی در پنجره **Awanwaves** به شکل زیر خواهد بود :



ارقام ثبت شده در فایل **LIS**. به قرار زیر هستند :

freq	voltage (1)	voltage (2)	current (r2)	current (c1)
1.00000k	1.0000	499.9975m	499.9975u	3.1416u
1.58489k	1.0000	499.9938m	499.9938u	4.9790u
2.51189k	1.0000	499.9844m	499.9844u	7.8911u
3.98107k	1.0000	499.9609m	499.9609u	12.5059u
6.30957k	1.0000	499.9018m	499.9018u	19.8182u
10.00000k	1.0000	499.7534m	499.7534u	31.4004u
15.84893k	1.0000	499.3814m	499.3814u	49.7293u
25.11886k	1.0000	498.4504m	498.4504u	78.6687u
39.81072k	1.0000	496.1347m	496.1347u	124.1022u
63.09573k	1.0000	490.4574m	490.4574u	194.4380u
100.00000k	1.0000	477.0141m	477.0141u	299.7168u
158.48932k	1.0000	447.5874m	447.5874u	445.7154u
251.18864k	1.0000	392.5065m	392.5065u	619.4792u
398.10717k	1.0000	312.2423m	312.2423u	781.0371u
630.95734k	1.0000	225.2079m	225.2079u	892.8189u
1.00000x	1.0000	151.6572m	151.6572u	952.8905u

تجزیه و تحلیل گذرا (.TRAN) :

در این تجزیه و تحلیل پاسخ گذرای یک مدار در حوزه زمان برای یک ورودی معین در حوزه زمان رسم و محاسبه می شود . جهت اجرای این تجزیه و تحلیل مدار باید دارای حداقل یک ورودی سینوسی ، پالسی ، نمایی یا پاره خط باشد .
دستور برنامه دارای یکی از دو فرم زیر است :

```
.TRAN tstep tstop <UIC>  
.TRAN tstart tstop tstep <UIC>
```


پارامترها کماکان واضح می باشند . **UIC** از حروف کلمات **Use Initial Condition** گرفته شده است . در صورتیکه سلف ها و خازنها دارای شرایط اولیه باشند ، جهت استفاده از آنها در این تجزیه و تحلیل باید کلمه **UIC** قید شود .
برنامه زیر را در نظر بگیرید :

MOS Inverter Circuit


```
.OPTION LIST NODE POST  
.MODEL PCH PMOS LEVEL = 1  
.MODEL NCH NMOS LEVEL = 1
```

```
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N  
M1 OUT IN VCC VCC PCH L = 1U W = 20U  
M2 OUT IN 0 0 NCH L = 1U W = 20U  
VCC VCC 0 5  
CLOAD OUT 0 .75P
```

```
.TRAN 200P 20N  
.END
```



شکل موجهای خروجی و ورودی به شکل زیرند :



دستور PRINT :

نتایج تجزیه و تحلیل های مختلف را می توان با کمک دستور PRINT بصورت عددی یا گرافیکی بدست آورد . شکل کلی آن به صورت زیر است :

متغیر های خروجی type PRINT .

type نشان دهنده نوع تحلیلی است که متغیر های آن به خروجی انتقال می یابند .

مثال :

```
.PRINT DC      V(1)  I(R2)  V(3,5)
.PRINT TRAN  V(in) V(out)
```

می توان در دستور بجای متغیرهای خروجی ، یک یا چند پارامتر دیگر را بصورت تابعی از ولتاژ و جریانهای نقاط مدار تعریف کرد . این پارامترها پس از تحلیل مدار به نام Params ظاهر می گردند . با کلیک کردن روی آنها می توان شکل موج آنها را دید . جهت تعریف توابع در دستور PRINT از عبارت ('PAR') استفاده می گردد .

مثال :

```
.PRINT GAIN=PAR('V(out)/V(in)')
```

دستور PROBE :

جهت دیدن شکل موجهای خواسته شده در تجزیه و تحلیل های AC ، DC و TRAN از این دستور استفاده می نمایم . در صورتیکه این دستور بدون پارامتری در برنامه گنجانده شود ، همه شکل موجها از متغیرهای مختلف نشان داده خواهند شد و در صورتیکه متغیرهای خاصی به عنوان پارامتر جلوی دستور ذکر شوند ، شکل موج مربوط به آنها نشان داده می شوند . استفاده از این دستور نیز کاملاً شبیه دستور PRINT می باشد .

دستور .PARAM :

با این دستور می توان به یک پارامتر ، عدد خاصی را تخصیص داد و سپس از این پارامتر به عنوان عدد ثابت در برنامه استفاده نمود .

مثال :

```
.PARAM P1=5n P2=7u  
.PARAM P3='2*20u'
```

در هر نقطه از برنامه که از " استفاده شود ، برنامه ابتدا مقدار درون آنرا محاسبه و سپس عمل تخصیص را انجام خواهد داد . در مثال اخیر پارامتر P3 حاوی مقدار 40u خواهد بود .

دستور .LIB :

با این دستور فایل متنی حاوی پارامتر های یک یا چند مدل ساخته می شوند و می توان در برنامه های آتی با فراخوانی این فایل ، از مدل های ذخیره شده در آن استفاده نمود .

```
.LIB entryname1  
entryname1 پارامتر های مدل  
.END entryname1  
entryname2 پارامتر های مدل  
.END entryname2  
.  
.
```



در برنامه جهت فراخوانی یک فایل متنی حاوی مدل های مختلف از دستورات زیر

استفاده می شود :

```
.INCLUDE '<path>\filename'  
.LIB '<path>\filename' <entryname>
```

دستور .TEMP :

جهت تحلیل مدار در درجه حرارتی مشخص بر حسب درجه سانتیگراد این دستور استفاده می شود . در صورت استفاده از چند درجه حرارت ، مدار به ازای هر درجه به طور مجزا تحلیل خواهد شد . مثال :

```
.TEMP 25 50 75
```